

ALGUNAS FUNCIONES ESTADÍSTICAS DE EXCEL

PROMEDIO	2
MEDIA.GEOM	4
MEDIA.ARMO	5
MODA	7
MEDIANA	8
PERCENTIL	9
RANGO.PERCENTIL	11
CUARTIL	13
VARP	15
VAR	17
DESVESTP	18
DESVEST	20
COEFICIENTE.ASIMETRIA	21
CURTOSIS	23
COVAR	25
COEF.DE.CORREL	26
PEARSON	26
COEFICIENTE.R2	28
PENDIENTE	29
INTERSECCION.EJE	31
DISTR.BINOM	32
POISSON	34
DISTR.HIPERGEOM	35
NEGBINOMDIST	37

Otras funciones de Excel.

CONTAR	39
MIN	40
MAX	42
SUMA	43
PRODUCTO	44
RAIZ	45
POTENCIA	46

Función PROMEDIO

Descripción

Devuelve el promedio (**media aritmética**) de los argumentos. Por ejemplo, si el rango A1:A20 contiene números, la fórmula **=PROMEDIO(A1:A20)** devuelve el promedio de dichos números.

Sintaxis

PROMEDIO(número1;[número2];...)

La sintaxis de la función **PROMEDIO** tiene los siguientes argumentos:

- **número1** (Obligatorio). El primer número, referencia de celda o rango para el que desea el promedio.
- **número2** , ... (Opcionales). Números, referencias de celda o rangos adicionales para los que desea el promedio, hasta un máximo de 255.

Observaciones

- La función PROMEDIO es la media aritmética:

$$\bar{x} = \frac{1}{n} \sum_{i=1}^n x_i$$

donde n es el número de datos y x_i representa los datos sobre los que se calcula la media aritmética.

- Los argumentos pueden ser números o nombres, rangos o referencias de celda que contengan números.
- Si el argumento de un rango o celda de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se puedan traducir a números provocan errores.
- Cuando esté calculando el promedio de celdas, tenga en cuenta la diferencia existente entre las celdas vacías y las que contienen el valor cero, las celdas vacías no se tienen en cuenta, pero sí los valores cero.

Ejemplos

	A	B	C	D
1	Datos			
2	10	15	32	
3	7	19	3	
4	9	6	7	
5	27	21	1	
6	2	0	15	

Fórmula	Descripción	Resultado
=PROMEDIO(A2:A6)	Promedio de los números en las celdas A2 a A6 {10, 7, 9, 27, 2}.	11
=PROMEDIO(A2:A7)	Promedio de los números en las celdas A2 a A7. El resultado coincide con el ejemplo anterior pues A7 está vacía.	11
=PROMEDIO(A2:C2)	Promedio de los números en las celdas A2 a C2 {10, 15, 32}.	19
=PROMEDIO(A2:C4)	Promedio de los números en las celdas A2, B2, C2, A3, B3, C3, A4, B4 y C4, es decir, de los datos de la matriz cuyas primera y última celdas (leída de izquierda a derecha y de arriba a abajo) son A2 y C4 {10, 15, 32, 7, 19, 3, 9, 6, 7} .	12
=PROMEDIO(A1:D4)	Promedio de los números en las celdas A1, B1, C1, D1, A2, B2, C2, D2, A3, B3, C3, D3, A4, B4, C4 y D4, es decir, de los datos de la matriz cuyas primera y última celdas (leída de izquierda a derecha y de arriba a abajo) son A1 y D4. El resultado coincide con el ejemplo anterior pues A1 contiene texto y B1, C1, D1, D2, D3 y D4 están vacías.	12
=PROMEDIO(A2:A6;5)	Promedio de los números en las celdas A2 a A6 y el número 5.	10
=PROMEDIO(A2:A6;C6;B5)	Promedio de los números en las celdas A2 a A6 y las celdas C6 y B5.	13

Normalmente los datos estarán en la misma columna (o en la misma fila), como en los tres primeros ejemplos.

Función MEDIA.GEOM

Descripción

Devuelve la media geométrica de una matriz o de un rango de datos positivos.

Por ejemplo, es posible utilizar la función MEDIA.GEOM para calcular la tasa de crecimiento medio, dado un crecimiento variable a lo largo del tiempo.

Sintaxis

MEDIA.GEOM(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si uno de los puntos de datos ≤ 0 , MEDIA.GEOM devuelve el valor de error #¡NUM!.
- La expresión para la media geométrica (MEDIA.GEOM) es:

$$G = \sqrt[n]{x_1 x_2 \dots x_n}$$

donde n es el número de datos y x_i representa los datos sobre los que se calcula la media geométrica.

Ejemplo

	A	B
1	Datos	
2	4	
3	5	
4	8	
5	7	
6	11	
7	4	
8	3	

Fórmula	Descripción	Resultado
=MEDIA.GEOM(A2:A8)	Media geométrica del conjunto de datos anterior	5,476987

Función MEDIA.ARMO

Descripción

Devuelve la media armónica de un conjunto de datos. La media armónica es la inversa de la media aritmética de los valores recíprocos o inversos.

Sintaxis

MEDIA.ARMO(número1;número2;...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- La media armónica es siempre inferior a la media geométrica, que a su vez es siempre inferior a la media aritmética.
- Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si uno de los puntos de datos ≤ 0 , MEDIA.ARMO devuelve el valor de error #¡NUM!.
- La expresión para la media armónica H (MEDIA.ARMO) es:

$$H = \left(\frac{1}{n} \sum_{i=1}^n \frac{1}{x_i} \right)^{-1}$$

donde n es el número de datos y x_i representa los datos sobre los que se calcula la media armónica.

Ejemplo

	A	B
1	Datos	
2	4	
3	5	
4	8	
5	7	
6	11	
7	4	
8	3	

Fórmula	Descripción	Resultado
=MEDIA.ARMO(A2:A8)	Media armónica del conjunto de datos anterior	5,028376

Función MODA

Descripción

Devuelve el valor que se repite con más frecuencia en una matriz o rango de datos. Por ejemplo, la moda de 2, 3, 3, 5, 7 y 10 es 3.

Sintaxis

MODA(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si el conjunto de datos no contiene datos duplicados, MODA devuelve el valor de error #N/A.
- Si existe más de una moda devuelve el valor de una de las modas.

Ejemplos

	A	B
1	Datos	Datos
2	5,6	2
3	4	3
4	4	3
5	3	4
6	2	4
7	4	5

Fórmula	Descripción	Resultado
=MODA(A2:A7)	Moda, es decir, número que se repite con más frecuencia	4
=MODA(B2:B7)	Hay dos modas, devuelve el valor de la primera que encuentra	3

Función MEDIANA

Descripción

Devuelve la mediana de los números dados. La mediana es el número que se encuentra justo en la mitad de un conjunto de números ordenados. Por ejemplo, la mediana de 2, 3, 3, 5, 7 y 10 es 4.

Sintaxis

MEDIANA(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- Si la cantidad de números en el conjunto es par, MEDIANA calcula el promedio de los números centrales. Vea la segunda fórmula del ejemplo.
 - Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
 - Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
 - Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
 - Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
-

Ejemplos

	A	B
1	Datos	
2	1	
3	2	
4	3	
5	4	
6	5	
7	6	

Fórmula	Descripción	Resultado
=MEDIANA(A2:A6)	La mediana de los 5 primeros números de la lista anterior	3
=MEDIANA(A2:A7)	La mediana de todos los números de la lista es el promedio de 3 y 4	3,5

Función PERCENTIL

Descripción

Devuelve el k-ésimo percentil de los valores de un rango.

Esta función puede permitir establecer un umbral de aceptación. Por ejemplo, se quiere examinar sólo a los candidatos cuya calificación sea superior al percentil 90.

Sintaxis

PERCENTIL(matriz;k)

matriz es la matriz o rango de datos sobre los que se quiere definir la posición relativa.

k es el rango u orden del percentil en el intervalo de 0 a 1, inclusive.

Observaciones

- En las distribuciones de variables discretas (datos aislados) **no hay consenso sobre la forma de calcular los percentiles**, existiendo en la literatura científica nueve métodos diferentes, que conducen a resultados diferentes. Por ello, al calcular cualquier percentil por medio de software o manualmente, es básico saber e indicar el método utilizado. **Excel no utiliza el mismo método que hemos estudiado en clase (TC1).**
- Si el argumento **matriz** está vacío o contiene más de 8.191 puntos de datos, PERCENTIL devuelve el valor de error #¡NUM!
- Si el argumento **k** no es numérico, PERCENTIL devuelve el valor de error #¡VALOR!
- Si **k** es < 0 o si **k** > 1 , PERCENTIL devuelve el valor de error #¡NUM!
- Si **k** no es un múltiplo de $1/(n - 1)$, donde n es el número de datos, la función PERCENTIL interpola para determinar el valor del k-ésimo percentil (obsérvese el segundo y quinto ejemplos).

Ejemplos

	A	B
1	Datos	Datos
2	2	1
3	4	1
4	5	2
5	6	2
6	8	2
7	9	4
8	10	4
9	12	5
10	15	5
11	16	5
12	20	7

Se han ordenado los datos de menor a mayor para facilitar la comprensión del cálculo de los percentiles.

Fórmula	Descripción	Resultado
=PERCENTIL(A2:A12;0,30)	$n=11$ datos. $1/(n-1)=0,10$. 0,30 es múltiplo de 0,10. El percentil 30 de la lista A2:A12 es 6 pues de los 10 datos (exceptuado él mismo) hay 3 menores que él, es decir $3/10=0,30$ de las observaciones son menores que 6.	6

=PERCENTIL(A2:A12;0,36)	El percentil 30 es 6, el percentil 40 es 8, interpolando 36 entre 30 y 40, obtenemos el percentil 36 $\frac{36-30}{40-30} = \frac{P_{36}-6}{8-6}$	7,2
=PERCENTIL(B2:B12;0,20)	$n=11$ datos. $1/(n-1)=0,10$. 0,20 es múltiplo de 0,10. Cuando hay datos repetidos como en la columna B, se razona de la siguiente forma: Hay dos datos por debajo (menores) del primer 2 de la lista, lo que representa $2/(n-1)=0,20$ de las observaciones.	2
=PERCENTIL(B2:B12;0,40)	Hay cuatro datos por debajo del último 2 de la lista, lo que representa $4/(n-1)=0,40$ de las observaciones.	2
=PERCENTIL(B2:B12;0,48)	El percentil 40 es 2, el percentil 50 es 4, interpolando 48 entre 40 y 50, obtenemos el percentil 48 $\frac{48-40}{50-40} = \frac{P_{48}-2}{4-2}$	3,6

Excel no utiliza el mismo método que hemos estudiado en clase (TC1).

Función RANGO.PERCENTIL

Descripción

Devuelve el rango de un valor en un conjunto de datos como el porcentaje del conjunto de datos menores a él.

Esta función le permite evaluar la posición relativa de un valor en un conjunto de datos. Por ejemplo, puede utilizar RANGO.PERCENTIL para evaluar la posición del resultado de una prueba de aptitud entre los resultados de la prueba.

Sintaxis

RANGO.PERCENTIL(matriz;x;cifra_significativa)

matriz es la matriz o rango de datos con valores numéricos en los que se define la posición relativa.

x es el valor cuyo rango percentil desea conocer.

cifra_significativa es un valor opcional que identifica el número de cifras significativas para el valor del porcentaje devuelto. Si se omite este argumento, RANGO.PERCENTIL utiliza tres dígitos (0,xxx).

Observaciones


- En las distribuciones de variables discretas (datos aislados) **no hay consenso sobre la forma de calcular los percentiles**, existiendo en la literatura científica nueve métodos diferentes, que conducen a resultados diferentes. Por ello, al calcular el rango de cualquier percentil por medio de software o manualmente, es básico saber e indicar el método utilizado. **Excel no utiliza el mismo método que hemos estudiado en clase (TC1)**.
- Si el argumento **matriz** está vacío, RANGO.PERCENTIL devuelve el valor de error #¡NUM!.
- Si el argumento **cifra_significativa** < 1, RANGO.PERCENTIL devuelve el valor de error #¡NUM!.
- Si el argumento **x** no coincide con uno de los valores del argumento matriz, RANGO.PERCENTIL interpola para devolver la jerarquía de porcentaje correcta (obsérvese el segundo y cuarto ejemplos).

Ejemplos

	A	B
1	Datos	Datos
2	2	1
3	4	1
4	5	2
5	6	2
6	8	2
7	9	4
8	10	4
9	12	5
10	15	5
11	16	5
12	20	7

Fórmula	Descripción	Resultado
=RANGO.PERCENTIL(A2:A12;15)	El rango percentil de 15 en la lista anterior (debido a que 8 valores del conjunto son menores que 15, y 2 son mayores que 15; $8/(8+2)=0,80$).	0,80

=RANGO.PERCENTIL(A2:A12;15,6)	El rango percentil de 15 es 0,80. El rango percentil de 16 es 0,90, interpolando 15,6 entre 15 y 16 obtenemos que el rango percentil de 15,6 es 0,86 $\frac{\alpha - 80}{90 - 80} = \frac{15,6 - 15}{16 - 15}$	0,86
=RANGO.PERCENTIL(B2:B12;4)	El rango percentil de 4 en la lista de la columna B es 5/10=0,50 pues hay 5 valores menores que 4 (5/10=0,50).	0,50
=RANGO.PERCENTIL(B2:B12;4,3)	El rango percentil del último 4 en la lista de la columna B es 6/10=0,60, el rango percentil del primer 5 es 7/10=0,70. Interpolando 4,3 entre 4 y 5 obtenemos que el rango percentil de 4,3 es 0,63 $\frac{\alpha - 60}{70 - 60} = \frac{4,3 - 4}{5 - 4}$	0,63

Excel no utiliza el mismo método que hemos estudiado en clase (TC1). Para ver el número como porcentaje, seleccione la celda y, a continuación, en la ficha **Hoja**, en el grupo **Número**, haga clic en **Porcentaje** .

Función CUARTIL

Descripción

Devuelve el cuartil de un conjunto de datos.

Los cuartiles se usan con frecuencia en los datos de encuestas para dividir las poblaciones en grupos. Por ejemplo, puede utilizar la función CUARTIL para determinar el 25 por ciento de ingresos más altos en una población.

Sintaxis

CUARTIL(matriz;cuartil)

matriz es la matriz o rango de celdas de valores numéricos cuyo cuartil desea obtener.

cuartil indica el valor que se devolverá.

SI CUARTIL ES IGUAL A	LA FUNCIÓN CUARTIL DEVUELVE
0	Valor mínimo
1	El primer cuartil (percentil 25)
2	El valor de la mediana (percentil 50)
3	El tercer cuartil (percentil 75)
4	Valor máximo

Observaciones

- En las distribuciones de variables discretas (datos aislados) **no hay consenso sobre la forma de calcular los cuartiles** (casos particulares de percentiles), existiendo en la literatura científica nueve métodos diferentes, que conducen a resultados diferentes. Por ello, al calcular cualquier cuartil por medio de software o manualmente, es básico saber e indicar el método utilizado. **Excel no utiliza el mismo método que hemos estudiado en clase (TC1).**
- Si el argumento **matriz** está vacío, CUARTIL devuelve el valor de error #¡NUM!
- Si el argumento **cuartil** no es un número entero, se trunca.
- Si **cuartil** < 0 o si **cuartil** > 4, CUARTIL devuelve el valor de error #¡NUM!
- Las funciones MIN, MEDIANA y MAX devuelven el mismo valor que CUARTIL cuando el argumento **cuartil** es igual a 0 (cero), 2 y 4 respectivamente.

Ejemplo

	A	B
1	Datos	
2	1	
3	2	
4	4	
5	7	
6	8	
7	9	
8	10	
9	12	

Fórmula	Descripción	Resultado
=CUARTIL(A2:A9;1)	Primer cuartil (percentil 25) de los datos anteriores	3,5

Excel no utiliza el mismo método que hemos estudiado en clase (TC1).

Función VARP

Descripción

Calcula la varianza de un conjunto de datos.

Sintaxis

VARP(número1;[número2]; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- VARP parte de la hipótesis de que los argumentos representan la población total (todos los datos sobre los que se quiere calcular la varianza). Si los datos representan una parte (muestra) de la población, utilice la función VAR para estimar la varianza de la población a partir de los datos de una muestra. **En clase (TC1) se ha supuesto siempre que se trabaja con todos los datos sobre los que queremos calcular la varianza, por lo que utilizaremos VARP para el cálculo de esta.**
- Los argumentos pueden ser números o nombres, rangos o referencias de celda que contengan números.
- Si el argumento de un rango o celda de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- La expresión de VARP es:

$$S^2 = \frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^2$$

Donde \bar{x} es la media aritmética (PROMEDIO), n es el número de datos y x_i representa los datos sobre los que se calcula la varianza.

Ejemplos

	A	B	C	D
1	Datos			
2	10	15	32	
3	7	19	3	
4	9	6	7	
5	27	21	1	
6	2	0	15	

Fórmula	Descripción	Resultado
=VARP(A2:A6)	Varianza de los números en las celdas A2 a A6 {10, 7, 9, 27, 2}.	71,6
=VARP(A2:A7)	Varianza de los números en las celdas A2 a A7. El resultado coincide con el ejemplo anterior pues A7 está vacía.	71,6
=VARP(A2:C2)	Varianza de los números en las celdas A2 a C2 {10, 15, 32}.	88,66667
=VARP(A2:C4)	Varianza de los números en las celdas A2, B2, C2, A3, B3, C3, A4, B4 y C4, es decir, de los datos de la matriz cuyas primera y última celdas (leída de izquierda a derecha y de arriba a abajo) son A2 y C4 {10, 15, 32, 7, 19, 3, 9, 6, 7}.	70,88889
=VARP(A1:D4)	Varianza de los números en las celdas A1, B1, C1, D1, A2, B2, C2, D2, A3, B3, C3, D3, A4, B4, C4 y D4, es decir, de los datos de la matriz cuyas primera y última celdas (leída de izquierda a derecha y de arriba a abajo) son A1 y D4. El resultado coincide con el ejemplo anterior pues A1 contiene texto y B1, C1, D1, D2, D3 y D4 están vacías.	70,88889
=VARP(A2:A6;5)	Varianza de los números en las celdas A2 a A6 y el número 5.	64,66667
=VARP(A2:A6;C6;B5)	Varianza de los números en las celdas A2 a A6 y las celdas C6 y B5.	63,71429

Normalmente los datos estarán en la misma columna (o en la misma fila), como en los tres primeros ejemplos.

Función VAR

Descripción

Estima la varianza de una población a partir de los datos de una muestra. Este estadístico es conocido como **cuasivarianza**.

Sintaxis

VAR(número1;[número2]; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

En todo es igual a la función VARP salvo que en la expresión de VAR se divide por $n-1$:

$$S_{n-1}^2 = \frac{1}{n-1} \sum_{i=1}^n (x_i - \bar{x})^2$$

Ambas funciones coinciden cuando $n \rightarrow \infty$ (en la práctica cuando n es suficientemente grande).

Ejemplo

	A
1	Datos
2	10
3	7
4	9
5	27
6	2

Fórmula	Descripción	Resultado
=VAR(A2:A6)	Cuasivarianza de los números en las celdas A2 a A6. Como puede observarse no coincide con el valor obtenido en el primer ejemplo de la función VARP sobre los mismos datos.	89,5

Función DESVESTP

Descripción

Calcula la desviación estándar (o desviación típica) de un conjunto de datos.

Sintaxis

DESVESTP(número1; [número2]; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- DESVESTP parte de la hipótesis de que los argumentos representan la población total (todos los datos sobre los que se quiere calcular la desviación típica). Si los datos representan una parte (muestra) de la población, utilice la función DESVEST para estimar la desviación típica de la población a partir de los datos de la muestra. **En clase (TC1) se ha supuesto siempre que se trabaja con todos los datos sobre los que queremos calcular la desviación típica, por lo que utilizaremos DESVESTP para el cálculo de esta.**
- Los argumentos pueden ser números o nombres, rangos o referencias de celda que contengan números.
- Si el argumento de un rango o celda de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- DESVESTP utiliza la expresión siguiente:

$$S = \sqrt{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^2}$$

Donde \bar{x} es la media aritmética de la muestra (PROMEDIO), n es el tamaño de la muestra y x_i representa los datos sobre los que se calcula la desviación típica.

Ejemplos

	A	B	C	D
1	Datos			
2	10	15	32	
3	7	19	3	
4	9	6	7	
5	27	21	1	
6	2	0	15	

Fórmula	Descripción	Resultado
=DESVESTP(A2:A6)	Desviación típica de los números en las celdas A2 a A6 {10, 7, 9, 27, 2}.	8,46168
=DESVESTP(A2:A7)	Desviación típica de los números en las celdas A2 a A7. El resultado coincide con el ejemplo anterior pues A7 está vacía.	8,46168
=DESVESTP(A2:C2)	Desviación típica de los números en las celdas A2 a C2 {10, 15, 32}.	9,41630
=DESVESTP(A2:C4)	Desviación típica de los números en las celdas A2, B2, C2, A3, B3, C3, A4, B4 y C4, es decir, de los datos de la matriz cuyas primera y última celdas (leída de izquierda a derecha y de arriba a abajo) son A2 y C4 {10, 15, 32, 7, 19, 3, 9, 6, 7}.	8,41955
=DESVESTP(A1:D4)	Desviación típica de los números en las celdas A1, B1, C1, D1, A2, B2, C2, D2, A3, B3, C3, D3, A4, B4, C4 y D4, es decir, de los datos de la matriz cuyas primera y última celdas (leída de izquierda a derecha y de arriba a abajo) son A1 y D4. El resultado coincide con el ejemplo anterior pues A1 contiene texto y B1, C1, D1, D2, D3 y D4 están vacías.	8,41955
=DESVESTP(A2:A6;5)	Desviación típica de los números en las celdas A2 a A6 y el número 5.	8,04156
=DESVESTP(A2:A6;C6;B5)	Desviación típica de los números en las celdas A2 a A6 y las celdas C6 y B5.	7,98212

Normalmente los datos estarán en la misma columna (o en la misma fila), como en los tres primeros ejemplos.

Función DESVEST

Descripción

Estima la desviación típica (o desviación estándar) de una población a partir de los datos de una muestra. Este estadístico es conocido como **cuasidesviación típica**.

Sintaxis

DESVEST(número1;[número2]; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

En todo es igual a la función DESVESTP salvo que en la expresión de DESVEST se divide por $n-1$:

$$S_{n-1} = \sqrt{\frac{1}{n-1} \sum_{i=1}^n (x_i - \bar{x})^2}$$

Ambas funciones coinciden cuando $n \rightarrow \infty$ (en la práctica cuando n es suficientemente grande).

Ejemplo

	A
1	Datos
2	10
3	7
4	9
5	27
6	2

Fórmula	Descripción	Resultado
=DESVEST(A2:A6)	Cuasi desviación típica de los números en las celdas A2 a A6. Como puede observarse no coincide con el valor obtenido en el primer ejemplo de la función DESVESTP sobre los mismos datos.	9,46044

Función COEFICIENTE.ASIMETRIA

Descripción

Estima la asimetría de una población con respecto a su media a partir de los datos de una muestra.

La asimetría positiva indica una distribución unilateral que se extiende hacia valores más positivos. La asimetría negativa indica una distribución unilateral que se extiende hacia valores más negativos.

Sintaxis

COEFICIENTE.ASIMETRIA(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si el número de puntos de datos es menor que tres o si la desviación estándar de la muestra es cero, COEFICIENTE.ASIMETRIA devuelve el valor de error #¡DIV/0!.
- La expresión de COEFICIENTE.ASIMETRIA es:

$$\frac{n}{(n-1)(n-2)} \sum_{i=1}^n \left(\frac{x_i - \bar{x}}{S_{n-1}} \right)^3$$

donde S_{n-1} es la estimación de la desviación típica de la población (DESVEST), \bar{x} es la media aritmética (PROMEDIO), n es el número de datos y x_i representa los datos sobre los que se calcula el coeficiente.

- **La expresión que utilizamos en TC1 para calcular el coeficiente de asimetría de Fisher es**

$$g_1 = \frac{m_3}{S^3} = \frac{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^3}{S^3} = \frac{1}{n} \sum_{i=1}^n \left(\frac{x_i - \bar{x}}{S} \right)^3$$

donde S es la desviación típica de los datos (DESVESTP), \bar{x} es la media aritmética (PROMEDIO), n es el número de datos y x_i representa los datos sobre los que se calcula el coeficiente, que coincide con el anterior cuando $n \rightarrow \infty$ (en la práctica cuando n es suficientemente grande). Como puede observarse en el ejemplo, donde $n=5$, la diferencia entre ambos coeficientes es notable.

Ejemplo

	A
1	Datos
2	10
3	7
4	9
5	27
6	2

Fórmula	Descripción	Resultado
=COEFICIENTE.ASIMETRIA(A2:A6)	Estima la asimetría de la población a partir de la muestra anterior.	1,62098
$g_1 = \frac{m_3}{S^3}$	Coefficiente de asimetría estudiado en TC1 sobre los datos anteriores.	1,08739

Función CURTOSIS

Descripción

Estima la curtosis o apuntamiento de una población a partir de los datos de una muestra.

La curtosis caracteriza la elevación o el achatamiento relativo de una distribución, comparada con la distribución normal. Una curtosis positiva indica una distribución relativamente elevada, mientras que una curtosis negativa indica una distribución relativamente plana.

Sintaxis

CURTOSIS(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función **PROMEDIO**.

Observaciones

- Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si existen menos de cuatro puntos de datos o si la desviación estándar de la muestra es igual a cero, la función CURTOSIS devuelve el valor de error #¡DIV/0!.
- La expresión de CURTOSIS es

$$\left\{ \frac{n(n+1)}{(n-1)(n-2)(n-3)} \sum_{i=1}^n \left(\frac{x_i - \bar{x}}{S_{n-1}} \right)^4 \right\} - \frac{3(n-1)^2}{(n-2)(n-3)}$$

donde S_{n-1} es la estimación de la desviación típica de la población (DESVEST), \bar{x} es la media aritmética (PROMEDIO), n es el número de datos y x_i representa los datos sobre los que se calcula el coeficiente.

- **La expresión que utilizamos en TC1 para calcular el coeficiente de curtosis de Fisher es**

$$g_2 = \frac{m_4}{S^4} - 3 = \frac{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^4}{S^4} - 3 = \frac{1}{n} \sum_{i=1}^n \left(\frac{x_i - \bar{x}}{S} \right)^4 - 3$$

donde S es la desviación típica de los datos (DESVESTP), \bar{x} es la media aritmética (PROMEDIO), n es el número de datos y x_i representa los datos sobre los que se calcula el coeficiente, que coincide con el anterior cuando $n \rightarrow \infty$ (en la práctica cuando n es suficientemente grande). Como puede observarse en el ejemplo, donde $n=5$, la diferencia entre ambos coeficientes es notable.

Ejemplo

	A
1	Datos
2	10
3	7
4	9
5	27
6	2

Fórmula	Descripción	Resultado
=CURTOSIS(A2:A6)	Estima el apuntamiento o curtosis de la población a partir de la muestra anterior.	3,29334
$g_2 = \frac{m_4}{S^4} - 3$	Coficiente de curtosis estudiado en TC1 sobre los datos anteriores.	-0,17666

Función COVAR

Descripción

Devuelve la covarianza, o promedio de los productos de las desviaciones para cada pareja de datos.

Utilice la covarianza para determinar las relaciones entre dos conjuntos de datos. Por ejemplo, puede investigar si unos ingresos más elevados se corresponden con niveles de estudios más altos.

Sintaxis

COVAR(matriz1;matriz2)

matriz1 es el primer rango de celdas de números.

matriz2 es el segundo rango de celdas de números.

Observaciones

- El orden de los argumentos **matriz1** y **matriz2** es indiferente. Si se escriben dichos argumentos en orden inverso se obtiene el mismo valor para la función COVAR.
- Los argumentos deben ser números o nombres, matrices o referencias que contengan números.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Si los argumentos **matriz1** y **matriz2** tienen distinto número de datos, COVAR devuelve el valor de error #N/A.
- Si los argumentos **matriz1** o **matriz2** están vacíos, COVAR devuelve el valor de error #DIV/0!
- La covarianza es:

$$S_{xy} = \frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})(y_i - \bar{y})$$

donde \bar{x} y \bar{y} son las medias aritméticas PROMEDIO(matriz1) y PROMEDIO(matriz2), n es el tamaño de la muestra y (x_i, y_i) son las parejas de datos sobre los que se calcula la covarianza.

Ejemplo

	A	B
1	Datos1	Datos2
2	3	9
3	2	7
4	4	12
5	5	15
6	6	17

Fórmula	Descripción	Resultado
=COVAR(A2:A6;B2:B6)	Covarianza, el promedio de los productos de las desviaciones para cada una de las parejas de datos anteriores	5,2

Función COEF.DE.CORREL o Función PEARSON

Descripción

Ambas funciones coinciden. Devuelven el coeficiente de correlación lineal de Pearson entre dos rangos de celdas definidos por los argumentos matriz1 y matriz2. El coeficiente de correlación lineal r es un índice adimensional acotado entre -1 y 1 que refleja el grado de dependencia lineal entre dos conjuntos de datos.

Use el coeficiente de correlación para determinar la relación entre dos variables. Por ejemplo, para examinar la relación entre la temperatura de una localidad y el uso del aire acondicionado.

Sintaxis

COEF.DE.CORREL(matriz1;matriz2)

PEARSON(matriz1;matriz2)

La sintaxis de estas funciones es análoga a la de la función **COVAR**.

Observaciones

- El orden de los argumentos **matriz1** y **matriz2** es indiferente. Si se escriben dichos argumentos en orden inverso se obtiene el mismo valor en ambas funciones.
- Los argumentos pueden ser números o nombres, matrices o referencias que contengan números.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Si los argumentos **matriz1** y **matriz2** tienen distinto número de datos, COEF.DE.CORREL (PEARSON) devuelve el valor de error #N/A.
- Si el argumento **matriz1** o **matriz2** está vacío, o si S (la desviación estándar) o sus valores son cero, COEF.DE.CORREL (PEARSON) devuelve el valor de error #¡DIV/0!
- La ecuación para el coeficiente de correlación lineal es:

$$r_{xy} = \frac{S_{xy}}{S_x S_y} = \frac{COVAR(matriz1; matriz2)}{DESVESTP(matriz1) \times DESVESTP(matriz2)} =$$

$$= \frac{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})(y_i - \bar{y})}{\sqrt{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^2} \sqrt{\frac{1}{n} \sum_{i=1}^n (y_i - \bar{y})^2}} = \frac{\sum_{i=1}^n (x_i - \bar{x})(y_i - \bar{y})}{\sqrt{\sum_{i=1}^n (x_i - \bar{x})^2} \sqrt{\sum_{i=1}^n (y_i - \bar{y})^2}}$$

donde \bar{x} e \bar{y} son las medias aritméticas PROMEDIO(matriz1) y PROMEDIO(matriz2), n es el tamaño de la muestra y (x_i, y_i) son las parejas de datos sobre los que se calcula el coeficiente.

Ejemplo

	A	B
1	Datos1	Datos2
2	3	9
3	2	7
4	4	12
5	5	15
6	6	17

Fórmula	Descripción	Resultado
=COEF.DE.CORREL(A2:A6;B2:B6)	Coeficiente de correlación de los dos conjuntos de datos anteriores	0,997054
=PEARSON(A2:A6;B2:B6)	Coeficiente de correlación de los dos conjuntos de datos anteriores	0,997054

Función COEFICIENTE.R2

Descripción

Devuelve el cuadrado del coeficiente de correlación lineal. Es un índice adimensional acotado entre 0 y 1. Este valor puede interpretarse como la proporción de la varianza de Y (*variabilidad o comportamiento de la variable Y*) que puede atribuirse a su relación con X.

Sintaxis

COEFICIENTE.R2(matriz1;matriz2)

La sintaxis de esta función es análoga a la de la función **COVAR**.

Observaciones

- El orden de los argumentos **matriz1** y **matriz2** es indiferente. Si se escriben dichos argumentos en orden inverso se obtiene el mismo valor en la función COEFICIENTE.R2.
- Los argumentos pueden ser números o nombres, matrices o referencias que contengan números.
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si los argumentos **matriz1** y **matriz2** están vacíos o tienen distinto número de datos, COEFICIENTE.R2 devuelve el valor de error #N/A.
- COEFICIENTE.R2 devuelve r^2 , que es el cuadrado del coeficiente de correlación lineal.

Ejemplo

	A	B
1	Datos1	Datos2
2	3	9
3	2	7
4	4	12
5	5	15
6	6	17
7		

Fórmula	Descripción	Resultado
=COEFICIENTE.R2(A2:A6;B2:B6)	Cuadrado del coeficiente de correlación lineal de los datos anteriores	0,99411765

Función PENDIENTE

Descripción

Devuelve la pendiente b de la recta de regresión de mínimos cuadrados de Y/X ($y = a + bx$).

Sintaxis

PENDIENTE(matriz_y;matriz_x)

matriz_y es una matriz o rango de celdas con los datos de la variable dependiente.

matriz_x es una matriz o rango de celdas con los datos de la variable independiente.

Observaciones

- El orden de los argumentos **matriz_y** y **matriz_x** no se puede cambiar como en las funciones COVAR, COEF.DE.CORREL, PEARSON y COEFICIENTE.R2. Si se escriben dichos argumentos en orden inverso se obtiene la pendiente b' de la recta de regresión de mínimos cuadrados de X/Y ($x=a'+b'y$).
- Los argumentos deben ser números o nombres, matrices o referencias que contengan números.

- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Si los argumentos **matriz_y** y **matriz_x** están vacíos o contienen un número diferente de datos, PENDIENTE devuelve el valor de error #N/A.
- La ecuación para la pendiente de la recta de regresión de mínimos cuadrados de Y/X es:

$$b = \frac{S_{xy}}{S_x^2} = \frac{COVAR(matriz_y; matriz_x)}{VARP(matriz_x)} = \frac{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})(y_i - \bar{y})}{\frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^2} = \frac{\sum_{i=1}^n (x_i - \bar{x})(y_i - \bar{y})}{\sum_{i=1}^n (x_i - \bar{x})^2}$$

donde \bar{x} e \bar{y} son las medias aritméticas PROMEDIO(matriz_x) y PROMEDIO(matriz_y), n es el tamaño de la muestra y (x_i, y_i) son los puntos sobre los que se ajusta la recta de regresión de mínimos cuadrados.

Ejemplo

	A	B
1	Datos x	Datos y
2	3	9
3	2	7
4	4	12
5	5	15
6	6	17
7		

Fórmula	Descripción	Resultado
=PENDIENTE(B2:B6;A2:A6)	Pendiente de la recta de regresión de mínimos cuadrados de Y/X para los datos anteriores	2,6

Función INTERSECCION.EJE

Descripción

Calcula el punto a en el que la recta de regresión de mínimos cuadrados de Y/X ($y = a + bx$) corta al eje Y .

Utilice la función INTERSECCION.EJE para determinar el valor de la variable dependiente cuando la variable independiente es igual a 0 (cero). Por ejemplo, puede emplear la función INTERSECCION.EJE para predecir la resistencia eléctrica de un metal a 0 °C si los puntos de datos se han tomado a temperatura ambiente o superior.

Sintaxis

INTERSECCION.EJE(matriz_y;matriz_x)

La sintaxis de esta función es análoga a la de la función **PENDIENTE**.

Observaciones

- El orden de los argumentos **matriz_y** y **matriz_x** no se puede cambiar como en las funciones COVAR, COEF.DE.CORREL, PEARSON y COEFICIENTE.R2. Si se escriben dichos argumentos en orden inverso se obtiene el punto a' en el que la recta de regresión de mínimos cuadrados de X/Y ($x=a'+b'y$) corta al eje X .
- Los argumentos deben ser números o nombres, matrices o referencias que contengan números.
- Si el argumento matricial o de referencia contiene texto, valores lógicos o celdas vacías, estos valores se pasan por alto; sin embargo, se incluirán las celdas con el valor cero.
- Si los argumentos **matriz_y** y **matriz_x** contienen un número diferente de puntos de datos o no contienen ninguno, INTERSECCION.EJE devuelve el valor de error #N/A.
- La ecuación que representa la intersección de la línea de regresión de mínimos cuadrados de Y/X , a , es:

$$a = \bar{y} - b\bar{x}$$

donde b es la **PENDIENTE(matriz_y;matriz_x)** y \bar{x} e \bar{y} son las medias aritméticas **PROMEDIO(matriz_x)** y **PROMEDIO(matriz_y)**.

Ejemplo

	A	B
1	Datos x	Datos y
2	3	9
3	2	7
4	4	12
5	5	15
6	6	17

Fórmula	Descripción	Resultado
=INTERSECCION.EJE(B2:B6;A2:A6)	El punto en el que la recta de regresión de mínimos cuadrados de Y/X interseca al eje Y	1,6

Función DISTR.BINOM

Descripción

Devuelve la probabilidad de una variable aleatoria discreta que sigue una distribución Binomial.

Utilice DISTR.BINOM en problemas con un número fijo de pruebas o ensayos, cuando los resultados de un ensayo son sólo éxito o fracaso, cuando los ensayos son independientes y cuando la probabilidad de éxito es constante durante todo el experimento. Por ejemplo, DISTR.BINOM puede calcular la probabilidad de que dos de los próximos tres bebés que nazcan sean hombres.

Sintaxis

DISTR.BINOM(núm_éxitos;ensayos;prob_éxito;acumulado)

núm_éxitos es el número de éxitos en los ensayos, **x**.

ensayos es el número de ensayos independientes, **n**.

prob_éxito es la probabilidad de éxito en cada ensayo, p .

acumulado es un valor lógico que determina la forma de la función. Si el argumento acumulado es **VERDADERO**, DISTR.BINOM devuelve la **función de distribución acumulada**, que es la probabilidad de que haya un número de éxitos menor o igual a x ; si es **FALSO**, devuelve la **función de masa de probabilidad**, que es la probabilidad de que haya x éxitos.

Observaciones

- Los argumentos **núm_éxitos** y **ensayos** se truncan a enteros.
- Si el argumento **núm_éxitos**, **ensayos** o **prob_éxito** no es numérico, DISTR.BINOM devuelve el valor de error #¡VALOR!
- Si el argumento **núm_éxitos** < 0 o si **núm_éxitos** > **ensayos**, DISTR.BINOM devuelve el valor de error #¡NUM!
- Si el argumento **prob_éxito** < 0 o si **prob_éxito** > 1, DISTR.BINOM devuelve el valor de error #¡NUM!
- La función de masa de probabilidad Binomial es (**acumulado**=FALSO):

$$p_x = P[X = x] = \binom{n}{x} p^x q^{n-x} \quad x = 0, \dots, n$$

- La función de distribución Binomial acumulada es (**acumulado**=VERDADERO):

$$F(x) = \sum_{y=0}^x p_y = \sum_{y=0}^x P[X = y] = \sum_{y=0}^x \binom{n}{y} p^y q^{n-y} \quad x = 0, \dots, n$$

Ejemplo

	A	B
1	Datos	Descripción
2	6	Número de éxitos en los ensayos
3	10	Número de ensayos independientes
4	0.5	Probabilidad de éxito en cada ensayo
5		

Fórmula	Descripción	Resultado
=DISTR.BINOM (A2;A3;A4;FALSO)	Probabilidad de que exactamente 6 de 10 ensayos sean éxito	0,205078
=DISTR.BINOM (A2;A3;A4;VERDADERO)	Probabilidad de que haya 6 éxitos o menos en los 10 ensayos	0,828125

Función POISSON

Descripción

Devuelve la probabilidad de una variable aleatoria discreta que sigue una distribución de Poisson.

Una de las aplicaciones comunes de la distribución de Poisson es calcular la probabilidad del número de eventos en un determinado período de tiempo, como por ejemplo, el número de automóviles que se presenta a una estación de peaje de una autopista en el intervalo de un minuto.

Sintaxis

POISSON(x;media;acumulado)

x es el número de eventos que se presentan.

media es el valor numérico esperado para la distribución de probabilidad de Poisson.

acumulado es un valor lógico que determina la forma de la distribución de probabilidad devuelta. Si el argumento acumulado es **VERDADERO**, POISSON devuelve la **función de distribución acumulada**, que es la probabilidad de que un evento aleatorio ocurra un número de veces menor o igual a **x**; si es **FALSO**, devuelve la **función de masa de probabilidad**, que es la probabilidad de que un evento aleatorio ocurra exactamente **x** veces.

Observaciones

- Si el argumento **x** no es un entero, se trunca.
- Si los argumentos **x** o **media** no son numéricos, POISSON devuelve el valor de error #¡VALOR!.
- Si **x** < 0, POISSON devuelve el valor de error #¡NUM!.
- Si **media** < 0, POISSON devuelve el valor de error #¡NUM!.
- La función de masa de probabilidad de Poisson es (**acumulado**=FALSO):

$$p_x = P[X = x] = \frac{e^{-\lambda} \lambda^x}{x!} \quad x = 0, 1, 2, \dots$$

- La función de distribución de Poisson acumulada es (**acumulado=VERDADERO**):

$$F(x) = \sum_{y=0}^x p_y = \sum_{y=0}^x P[X = y] = \sum_{y=0}^x \frac{e^{-\lambda} \lambda^y}{y!} \quad x = 0, 1, 2, \dots$$

Donde λ es la media o valor esperado de la distribución de probabilidad de Poisson.

Ejemplo

	A	B
1	Datos	Descripción
2	2	Número de eventos
3	5	Media o valor esperado
4		

Fórmula	Descripción	Resultado
=POISSON(A2;A3;VERDADERO)	Probabilidad de que se presenten 2 eventos o menos	0,124652
=POISSON(A2;A3;FALSO)	Probabilidad de que se presenten exactamente 2 eventos	0,084224

Función DISTR.HIPERGEOM

Descripción

Devuelve la probabilidad para una variable aleatoria discreta que sigue una distribución Hipergeométrica.

La función DISTR.HIPERGEOM devuelve la probabilidad de obtener un número determinado de "éxitos" en una muestra, conocidos el tamaño de la muestra, el número de éxitos en la población y el tamaño de la población. Utilice DISTR.HIPERGEOM en problemas con una población finita, donde cada observación sea un éxito o un fracaso, y donde cada subconjunto de un tamaño determinado pueda elegirse con la misma probabilidad.

Sintaxis

DISTR.HIPERGEOM(muestra_éxitos;tam_de_muestra;población_éxitos;tam_de_población)

muestra_éxitos es el número de éxitos en la muestra, **x**.

tam_de_muestra es el tamaño de la muestra, **n**.

población_éxitos es el número de éxitos en la población, **Np** (donde *p* es la probabilidad de éxito al observar un elemento de la población).

tam_de_población es el tamaño de la población, **N**.

Observaciones

- Todos los argumentos se truncan a enteros.
- Si uno de los argumentos no es numérico, DISTR.HIPERGEOM devuelve el valor de error #¡VALOR!
- Si el argumento **muestra_éxitos** < 0 o si **muestra_éxitos** es mayor que el menor de los números **tam_de_muestra** o **tam_de_población**, DISTR.HIPERGEOM devuelve el valor de error #¡NUM!
- Si el argumento **muestra_éxitos** es menor que el mayor número entre 0 y (**tam_de_muestra** - **tam_de_población** + **población_éxitos**), DISTR.HIPERGEOM devuelve el valor de error #¡NUM!
- Si el argumento **tam_de_muestra** ≤ 0 o si **tam_de_muestra** > **tam_de_población**, DISTR.HIPERGEOM devuelve el valor de error #¡NUM!
- Si el argumento **población_éxitos** ≤ 0 o si **población_éxitos** > **tam_de_población**, DISTR.HIPERGEOM devuelve el valor de error #¡NUM!
- Si el argumento **tam_de_población** ≤ 0, DISTR.HIPERGEOM devuelve el valor de error #¡NUM!
- La distribución de probabilidad Hipergeométrica está dada por:

$$p_x = P[X = x] = \frac{\binom{Np}{x} \binom{Nq}{n-x}}{\binom{N}{n}} \quad \max\{0, n-Nq\} \leq x \leq \min\{n, Np\}$$

donde:

x = muestra_éxitos

n = tam_de_muestra

Np = población_éxitos

Nq = *N* - *Np* = número de fracasos en la población, (*q* = 1 - *p*)

N = tam_de_población

La función DISTR.HIPERGEOM se utiliza en muestreos sin reemplazamiento, a partir de una población finita.

Ejemplo

Una caja de bombones contiene 20 piezas. Ocho de ellas contienen caramelo y las 12 restantes contienen nueces. Si una persona selecciona 4 bombones al azar, la siguiente función devuelve la probabilidad de que exactamente 1 contenga caramelo.

	A	B
1	Datos	Descripción
2	1	Número de éxitos en la muestra
3	4	Tamaño de la muestra
4	8	Número de éxitos en la población
5	20	Tamaño de la población

Fórmula	Descripción	Resultado
=DISTR.HIPERGEOM(A2;A3;A4;A5)	Distribución Hipergeométrica de la muestra y población anteriores	0,363261

Función NEGBINOMDIST

Descripción

Devuelve la distribución **Binomial negativa**. NEGBINOMDIST devuelve la probabilidad de obtener un **número de fracasos** igual al argumento **núm_fracasos** antes de lograr **el éxito determinado por el argumento núm_éxitos**, cuando la probabilidad de éxito, definida por el argumento **prob_éxito**, es **constante**. Se supone que los ensayos son independientes.

En el **caso particular** de que **núm_éxitos** sea **uno** tendremos la **distribución de probabilidad Geométrica**.

Por ejemplo, supongamos que necesita encontrar 10 personas que dispongan de excelentes reflejos y se sabe que la probabilidad de que un candidato tenga esta cualidad es 0,3. NEGBINOMDIST calcula la probabilidad de que entrevistará un número determinado de candidatos no calificados antes de encontrar los 10 candidatos buscados.

Sintaxis

NEGBINOMDIST(núm_fracasos;núm_éxitos;prob_éxito)

núm_fracasos es el número de fracasos.

núm_éxitos es el número límite de éxitos.

prob_éxito es la probabilidad de obtener un éxito.

Observaciones

- Los argumentos **núm_fracasos** y **núm_éxitos** se truncan a enteros.
- Si uno de los argumentos no es numérico, NEGBINOMDIST devuelve el valor de error #¡VALOR!
- Si el argumento **prob_éxito** < 0 o si **prob_éxito** > 1, NEGBINOMDIST devuelve el valor de error #¡NUM!
- Si los argumentos **núm_fracasos** < 0 o **núm_éxitos** < 1, la función NEGBINOMDIST devuelve el valor de error #¡NUM!
- La expresión para la **distribución de probabilidad Binomial negativa** es:

$$p_x = P[X = x] = \binom{x+r-1}{r-1} p^r q^x \quad x = 0, \dots$$

Donde, **x** es **núm_fracasos**, **r** es **núm_éxitos**, **p** es **prob_éxito** y $q=1-p$.

Ejemplo

	A	B
1	Datos	Descripción
2	10	Número de fracasos
3	5	Número de éxitos
4	0,25	Probabilidad de obtener un éxito

Fórmula	Descripción	Resultado
=NEGBINOMDIST(A2;A3;A4)	Distribución binomial negativa para los términos anteriores.	0,055049

NOTA: Si el número de éxitos es igual a uno se trata de la **distribución de probabilidad Geométrica**.

OTRAS FUNCIONES DE EXCEL

Aunque las siguientes funciones de Excel no son exclusivamente estadísticas, se han incluido debido a que son muy útiles en los cálculos habituales de Estadística Descriptiva.

Función CONTAR

Descripción

La función **CONTAR** cuenta la cantidad de celdas que contienen números. Por ejemplo, puede escribir la siguiente fórmula para contar los números en el rango A1:A20: =CONTAR(A1:A20). Si cinco de las celdas del rango contienen números, el resultado es 5.

Sintaxis

CONTAR(número 1; [número 2]; ...)

La sintaxis de esta función es análoga a la de la función PROMEDIO.

Los argumentos pueden contener o hacer referencia a una variedad de diferentes tipos de datos, pero sólo se cuentan los números.

Observaciones

- Se cuentan argumentos que son números, fechas o una representación de texto de los números (por ejemplo, un número entre comillas, como "1").
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- No se cuentan los argumentos que sean valores de error o texto que no se puedan traducir a números.
- Si un argumento es una matriz o una referencia, sólo se considerarán los números de esa matriz o referencia. No se cuentan celdas vacías, valores lógicos, texto o valores de error de la matriz o de la referencia.
- Si desea contar valores lógicos, texto o valores de error, use la función **CONTARA**.

- Si desea contar sólo números que cumplan con determinados criterios, use la función **CONTAR.SI** o la función **CONTAR.SI.CONJUNTO**.

Ejemplos

	A	B
1	Datos	
2	Ventas	
3	8/12/2008	
4		
5	19	
6	22,24	
7	VERDADERO	
8	#¡DIV/0!	

Fórmula	Descripción	Resultado
=CONTAR(A2:A8)	Cuenta la cantidad de celdas que contienen números en las celdas de A2 a A8.	3
=CONTAR(A5:A8)	Cuenta la cantidad de celdas que contienen números en las celdas de A5 a A8.	2
=CONTAR(A2:A8;2)	Cuenta la cantidad de celdas que contienen números en las celdas de A2 a A8 y el valor 2	4

Función MIN

Descripción

Devuelve el valor mínimo de un conjunto de valores.

Sintaxis

MIN(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función PROMEDIO.

Observaciones

- Los argumentos pueden ser números, o nombres, matrices o referencias que contengan números.
- Se tienen en cuenta los valores lógicos y las representaciones textuales de números escritos directamente en la lista de argumentos.
- Si el argumento es una matriz o una referencia, sólo se utilizarán los números contenidos en la matriz o en la referencia. Se pasarán por alto las celdas vacías, los valores lógicos o el texto contenidos en la matriz o en la referencia.
- Si los argumentos no contienen números, MIN devuelve 0.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.
- Si desea incluir valores lógicos y representaciones textuales de números en una referencia como parte del cálculo, utilice la función MINA.

Ejemplo

	A	B
1	Datos	
2	10	
3	7	
4	9	
5	27	
6	2	

Fórmula	Descripción	Resultado
=MIN(A2:A6)	El menor de los números anteriores	2
=MIN(A2:A6;0)	El menor de los números anteriores y 0	0

Función MAX

Descripción

Devuelve el valor máximo de un conjunto de valores.

Sintaxis

MAX(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función PROMEDIO.

Observaciones

Las mismas que para la función MIN, cambiando en la última observación MINA por MAXA

Ejemplo

	A	B
1	Datos	
2	10	
3	7	
4	9	
5	27	
6	2	

Fórmula	Descripción	Resultado
=MAX(A2:A6)	El mayor de los números anteriores	27
=MAX(A2:A6;40)	El mayor de los números anteriores y 40	40

Función SUMA

Descripción

Suma todos los números de un rango.

Sintaxis

SUMA(número1;número2; ...)

La sintaxis de esta función es análoga a la de la función PROMEDIO.

Observaciones

- Se tienen en cuenta los números, valores lógicos y representaciones textuales de números que se escriban directamente en la lista de argumentos. Vea los dos primeros ejemplos.
- Si un argumento es una matriz o una referencia, sólo se considerarán los números de esa matriz o referencia. Se pasarán por alto las celdas vacías, los valores lógicos o el texto contenidos en la matriz o en la referencia. Vea el tercer ejemplo a continuación.
- Los argumentos que sean valores de error o texto que no se pueda traducir a números provocan errores.

Ejemplos

	A	B
1	Datos	
2	-5	
3	15	
4	30	
5	'5	
6	VERDADERO	

Fórmula	Descripción	Resultado
=SUMA(3;2)	Suma 3 y 2	5
=SUMA("5";15;VERDADERO)	Suma 5, 15 y 1, ya que los valores de texto se traducen como números y el valor lógico VERDADERO se traduce como 1 .	21

=SUMA(A2:A6)	Suma los tres primeros números de la columna anterior. Puesto que los valores no numéricos de las referencias no se traducen, no se tienen en cuenta los valores de A5 y A6.	40
=SUMA(A2:A6;15)	Suma los tres primeros números de la columna anterior y 15	55
=SUMA(A5;A6;2)	Suma los valores de las dos últimas filas anteriores y 2. Puesto que los valores no numéricos de las referencias no se traducen, no se tienen en cuenta los valores de la columna anterior	2

Función PRODUCTO

Descripción

La función **PRODUCTO** multiplica todos los números proporcionados como argumentos y devuelve el producto. Por ejemplo, si las celdas A1 y A2 contienen números, puede usar la fórmula **=PRODUCTO(A1, A2)** para multiplicar los dos números. También puede realizar la misma operación con el operador matemático de multiplicación (*); por ejemplo, **=A1 * A2**.

La función **PRODUCTO** es útil cuando necesita multiplicar varias celdas. Por ejemplo, la fórmula **=PRODUCTO(A1:A3; C1:C3)** es equivalente a **=A1 * A2 * A3 * C1 * C2 * C3**.

Sintaxis

PRODUCTO(número 1; [número 2]; ...)

La sintaxis de esta función es análoga a la de la función PROMEDIO.

Si el argumento es una matriz o una referencia, sólo se multiplicarán los números de la matriz o referencia. Se omitirán las celdas vacías, los valores lógicos o el texto de la matriz o referencia.

Observaciones

Las mismas que para la función SUMA

Ejemplo

	A	B
1	Datos	
2	5	
3	15	
4	30	

Fórmula	Descripción	Resultado
=PRODUCTO(A2:A4)	Multiplica los números de las celdas A2 a A4.	2250
=PRODUCTO(A2:A4;2)	Multiplica los números de las celdas A2 a A4 y, a continuación, multiplica ese resultado por 2.	4500
=A2*A3*A4	Multiplica los números de las celdas A2 a A4 mediante operadores matemáticos en lugar de la función PRODUCTO .	2250

Función RAIZ

Descripción

Devuelve la raíz cuadrada de un número.

Sintaxis

RAIZ(número)

número es el número cuya raíz cuadrada desea obtener.

Observaciones

Si número es negativo, RAIZ devuelve el valor de error #¡NUM!

Ejemplo

	A	B
1	Datos	
2	-16	

Fórmula	Descripción	Resultado
=RAIZ(16)	Raíz cuadrada de 16	4
=RAIZ(A2)	Raíz cuadrada del número anterior. Puesto que el número es negativo, se devuelve un error	#¡NUM!
=RAIZ(ABS(A2))	Raíz cuadrada del valor absoluto del número anterior	4

Función POTENCIA

Descripción

Eleva un número a una potencia. También se puede utilizar el operador "^" en lugar de la función POTENCIA.

Sintaxis

POTENCIA(número 1; número 2)

número 1 es la base de la potencia.

número 2 es el exponente de la potencia.

Observaciones

- Si **número 2** es igual a $\frac{1}{2}$ la función **POTENCIA** coincide con la función **RAIZ** sobre **número 1**.
- Si **número 2** es igual a $\frac{1}{3}$, $\frac{1}{4}$, ... la función **POTENCIA** nos permite hallar las raíces de distintos ordenes sobre **número 1**.

Ejemplo

Fórmula	Descripción	Resultado
=POTENCIA(5;2)	Calcula el cuadrado de cinco	25
=5^3	Calcula el cubo de cinco	125
=POTENCIA(125;1/3)	Calcula la raíz cúbica de 125	5